Simulation of three phase rectifier




```
> restart;
```

Useful constant

```
> INT := Pi/6 ;
```

$$INT := \frac{1}{6}\,\pi$$

Upper voltage envelope. Phase are shifted by Pi/6 to have an easier function envelop

```
> fup := piecewise( t < 4*INT,  sin( t+INT ),
                    t < 8*INT,  sin( t+INT - 2*Pi/3 ),
                    t < 12*INT, sin( t+INT - 4*Pi/3) ) ;
```

$$fup := \begin{cases} \sin\left(t + \frac{1}{6}\,\pi\right) & t < \frac{2}{3}\,\pi \\ -\cos(t) & t < \frac{4}{3}\,\pi \\ \cos\left(t + \frac{1}{3}\,\pi\right) & t < 2\,\pi \end{cases}$$

Lower voltage envelope. Phase are shifter by Pi/6 to have an easier function envelop

```
> fdown := piecewise( t < 2*INT,  sin( t+INT - 2*Pi/3),
                      t < 6*INT,  sin( t+INT - 4*Pi/3),
                      t < 10*INT, sin( t+INT ),
                      t < 12*INT, sin( t+INT - 2*Pi/3) ) ;
```

$$fdown := \begin{cases} -\cos(t) & t < \frac{1}{3}\pi \\ \cos\left(t + \frac{1}{3}\pi\right) & t < \pi \\ \sin\left(t + \frac{1}{6}\pi\right) & t < \frac{5}{3}\pi \\ -\cos(t) & t < 2\pi \end{cases}$$

```
> with(plots):
```

Three phase plot

```
> A := plot( [sin(t+INT),sin(t+INT-2*Pi/3),sin(t+INT-4*Pi/3)], t=
  -2*Pi..2*Pi):
```

Plot upper envelope

```
> B := plot( fup(t),t=0..2*Pi,color=blue,thickness=3):
```

Plot lower envelope

```
> C := plot( subs(freq=1/Pi/2,fdown(t)),t=0..2*Pi,color=blue,
  thickness=3):
> display(A,B,C) ;
```



Compute Fourier coefficient of upper envelope

```
> au0 := simplify( int( fup,t=0..2*Pi)/Pi ) assuming freq > 0 ;
```

$$au0 := \frac{3\sqrt{3}}{\pi}$$

```
> auk1 := simplify( int( fup*cos(3*k*t),t=0..2*Pi)/Pi ) assuming
  freq > 0, k::integer ;
  auk2 := simplify( int( fup*cos((3*k+1)*t),t=0..2*Pi)/Pi )
  assuming freq > 0, k::integer ;
  auk3 := simplify( int( fup*cos((3*k+2)*t),t=0..2*Pi)/Pi )
  assuming freq > 0, k::integer ;
```

$$auk1 := -\frac{3\sqrt{3}}{\left(9k^2-1\right)\pi}$$

$$auk2 := 0$$

$$auk3 := 0$$

```
> buk1 := simplify( int( fup*sin(3*k*t),t=0..2*Pi)/Pi ) assuming
  freq > 0, k::integer ;
  buk2 := simplify( int( fup*sin((3*k+1)*t),t=0..2*Pi)/Pi )
  assuming freq > 0, k::integer ;
  buk3 := simplify( int( fup*sin((3*k+2)*t),t=0..2*Pi)/Pi )
  assuming freq > 0, k::integer ;
```

$$buk1 := 0$$

$$buk2 := 0$$

$$buk3 := 0$$

Compute Fourier coefficient of lower envelope

```
> al0 := simplify( int( fdown,t=0..2*Pi)/Pi ) assuming freq > 0 ;
```

$$al0 := -\frac{3\sqrt{3}}{\pi}$$

```
> alk1 := simplify( int( fdown*cos(3*k*t),t=0..2*Pi)/Pi ) assuming
  freq > 0, k::integer ;
  alk2 := simplify( int( fdown*cos((3*k+1)*t),t=0..2*Pi)/Pi )
  assuming freq > 0, k::integer ;
  alk3 := simplify( int( fdown*cos((3*k+2)*t),t=0..2*Pi)/Pi )
  assuming freq > 0, k::integer ;
```

$$alk1 := \frac{3\,(-1)^k\sqrt{3}}{\left(9\,k^2-1\right)\pi}$$

$$alk2 := 0$$

$$alk3 := 0$$

```
> blk1 := simplify( int( fdown*sin(3*k*t),t=0..2*Pi)/Pi ) assuming
  freq > 0, k::integer ;
  blk2 := simplify( int( fdown*sin((3*k+1)*t),t=0..2*Pi)/Pi )
  assuming freq > 0, k::integer ;
  blk3 := simplify( int( fdown*sin((3*k+2)*t),t=0..2*Pi)/Pi )
  assuming freq > 0, k::integer ;
```

$$blk1 := 0$$

$$blk2 := 0$$

$$blk3 := 0$$

```
> a0  := au0 - al0 ;
  a3k := simplify(auk1 - alk1) ;
  a6k := simplify(subs(k=2*k,auk1 - alk1)) assuming k::integer ;
```

$$a0 := \frac{6\sqrt{3}}{\pi}$$

$$a3k := -\frac{3\sqrt{3}\,\left(1+(-1)^k\right)}{\left(9\,k^2-1\right)\pi}$$

$$a6k := -\frac{6\sqrt{3}}{\left(36\,k^2-1\right)\pi}$$

First terms expansions

```
> ft0 := a0/2 ;
  ft1 := ft0 + subs(k=1,a6k)*cos(6*t) ;
  ft2 := ft1 + subs(k=2,a6k)*cos(12*t) ;
  ft3 := ft2 + subs(k=3,a6k)*cos(18*t) ;
```

$$ft0 := \frac{3\sqrt{3}}{\pi}$$

$$ft1 := \frac{3\sqrt{3}}{\pi} - \frac{6}{35}\,\frac{\sqrt{3}\cos(6\,t)}{\pi}$$

$$ft2 := \frac{3\sqrt{3}}{\pi} - \frac{6}{35}\,\frac{\sqrt{3}\cos(6\,t)}{\pi} - \frac{6}{143}\,\frac{\sqrt{3}\cos(12\,t)}{\pi}$$

$$ft3 := \frac{3\sqrt{3}}{\pi} - \frac{6}{35}\,\frac{\sqrt{3}\cos(6\,t)}{\pi} - \frac{6}{143}\,\frac{\sqrt{3}\cos(12\,t)}{\pi} - \frac{6}{323}\,\frac{\sqrt{3}\cos(18\,t)}{\pi}$$

```
> plot([fup(t)-fdown(t),ft1],t=0..2*Pi,y=0..2,thickness=[2,2]) ;
```




```
> plot([fup(t)-fdown(t),ft0,ft1,ft2,ft3],t=0..2*Pi,thickness=[3,2,
1,1]) ;
```



Ratio of continuous versus ripple coefficient

```
> a6k/(a0/2) ;
```

$$-\frac{2}{36\,k^2-1}$$

```
> -subs(k=1,a6k)/(a0/2) ;
  -subs(k=2,a6k)/(a0/2) ;
  -subs(k=3,a6k)/(a0/2) ;
  -subs(k=4,a6k)/(a0/2) ;
```

$$\frac{2}{35}$$

$$\frac{2}{143}$$

$$\frac{2}{323}$$

$$\frac{2}{575}$$

```
>
```